



HAL
open science

A surface of degree 24 with 1440 singularities of type D_{-4}
Cédric Bonnafé

► **To cite this version:**

| Cédric Bonnafé. A surface of degree 24 with 1440 singularities of type D_{-4} . 2018. hal-01773229v3

HAL Id: hal-01773229

<https://hal.science/hal-01773229v3>

Preprint submitted on 3 May 2018 (v3), last revised 3 Jul 2018 (v4)

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

A SURFACE OF DEGREE 24 WITH 1440 SINGULARITIES OF TYPE D_4

by

CÉDRIC BONNAFÉ

Abstract. — Using the invariant algebra of the complex reflection group denoted by G_{32} in the Shephard–Todd classification, we construct three irreducible surfaces in $\mathbf{P}^3(\mathbb{C})$ with many singularities: one of them has degree 24 and contains 1440 quotient singularities of type D_4 .

Let $\mu_{D_4}(d)$ denote the maximal number of quotient singularities of type D_4 that an irreducible projective surface of degree d in $\mathbf{P}^3(\mathbb{C})$ might have. Miyaoka [Miy] proved that

$$\mu_{D_4}(d) \leq \frac{16}{117} d(d-1)^2.$$

For $d = 8, 16$ or 24 , this reads

$$\mu_{D_4}(8) \leq 53, \quad \mu_{D_4}(16) \leq 492 \quad \text{and} \quad \mu_{D_4}(24) \leq 1736.$$

The main results of this note are that

$$(1) \quad \mu_{D_4}(8) \geq 44, \quad \mu_{D_4}(16) \geq 352 \quad \text{and} \quad \mu_{D_4}(24) \geq 1440$$

and that

$$(2) \quad \mu_{D_4}(8k) \geq 44k^3$$

for all $k \geq 1$. So, let $\mathbb{C}[x_1, x_2, x_3, x_4]$ be the polynomial ring over \mathbb{C} in 4 indeterminates with its usual grading and let $\mathbf{P}^3(\mathbb{C}) = \text{Proj}(\mathbb{C}[x_1, x_2, x_3, x_4])$ be the associated projective space of dimension 3. If $f \in \mathbb{C}[x_1, x_2, x_3, x_4]$ is homogeneous, we denote by $\mathcal{Z}(f)$ the projective hypersurface it defines, and by $\mathcal{Z}(f)_{\text{sing}}$ its reduced singular locus. If k is a natural number, we denote by $f[k]$ the homogeneous polynomial $f(x_1^k, x_2^k, x_3^k, x_4^k)$. We aim to prove the following result:

Theorem 1. *There exists a homogenous polynomial g of degree 8 such that:*

- (a) $\mathcal{Z}(g)$ is an irreducible surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with exactly 44 singular points which are all quotient singularities of type D_4 .
- (b) If $k \geq 1$, then $\mathcal{Z}(g[k])$ is an irreducible surface of degree $8k$, whose singular locus has dimension 0 and contains at least $44k^3$ quotient singularities of type D_4 .
- (c) $\mathcal{Z}(g[2])$ is an irreducible surface of degree 16 with exactly 472 singular points: 24 quotient singularities of type A_1 , 96 quotient singularities of type A_2 and 352 quotient singularities of type D_4 .
- (d) $\mathcal{Z}(g[3])$ is an irreducible surface of degree 24 in $\mathbf{P}^3(\mathbb{C})$ with exactly 1440 singular points which are all quotient singularities of type D_4 . The automorphism group of $\mathcal{Z}(g[3])$ contains at least 25920 elements and acts transitively on the 1440 singular points.

The proof relies on extensive MAGMA computations [Magma], and goes as follow. Let

$$\begin{aligned}
g = & x_1^8 - 6x_1^6x_2^2 - 60x_1^6x_2x_3 - 60x_1^6x_2x_4 - 6x_1^6x_3^2 + 60x_1^6x_3x_4 - 6x_1^6x_4^2 \\
& + 2240x_1^5x_2^2x_3 - 2240x_1^5x_2^2x_4 + 2240x_1^5x_2x_3^2 - 2240x_1^5x_2x_4^2 + 2240x_1^5x_3^2x_4 \\
& - 2240x_1^5x_3x_4^2 - 14x_1^4x_2^4 + 10180x_1^4x_2^3x_3 + 10180x_1^4x_2^3x_4 + 40412x_1^4x_2^2x_3^2 \\
& - 23440x_1^4x_2^2x_3x_4 + 40412x_1^4x_2^2x_4^2 + 10180x_1^4x_2x_3^3 + 23440x_1^4x_2x_3^2x_4 \\
& + 23440x_1^4x_2x_3x_4^2 + 10180x_1^4x_2x_4^3 - 14x_1^4x_3^4 - 10180x_1^4x_3^3x_4 + 40412x_1^4x_3^2x_4^2 \\
& - 10180x_1^4x_3x_4^3 - 14x_1^4x_4^4 + 10180x_1^3x_2^3x_3 - 10180x_1^3x_2^3x_4 + 111980x_1^3x_2^2x_3^2 \\
& - 111980x_1^3x_2^2x_4^2 + 111980x_1^3x_2x_3^3 - 111980x_1^3x_2x_4^3 + 10180x_1^3x_3^2x_4^2 - 10180x_1^3x_3x_4^3 \\
& + 10180x_1^3x_4^4 - 111980x_1^3x_3^2x_4^2 + 111980x_1^3x_3x_4^3 - 10180x_1^3x_4^4 - 6x_1^2x_2^6 \\
& + 2240x_1^2x_2^5x_3 + 2240x_1^2x_2^5x_4 + 40412x_1^2x_2^4x_3^2 + 23440x_1^2x_2^4x_3x_4 + 40412x_1^2x_2^4x_4^2 \\
& + 111980x_1^2x_2^3x_3^3 + 111980x_1^2x_2^3x_4^3 + 40412x_1^2x_2^2x_3^4 + 154704x_1^2x_2^2x_3^2x_4^2 + 40412x_1^2x_2^2x_4^3 \\
& + 2240x_1^2x_2x_3^5 - 23440x_1^2x_2x_3^4x_4 - 23440x_1^2x_2x_3x_4^4 + 2240x_1^2x_2x_4^5 - 6x_1^2x_3^6 \\
& - 2240x_1^2x_3^5x_4 + 40412x_1^2x_3^4x_4^2 - 111980x_1^2x_3^3x_4^3 + 40412x_1^2x_3^2x_4^4 - 2240x_1^2x_3x_4^5 \\
& - 6x_1^2x_4^6 - 60x_1x_2^6x_3 + 60x_1x_2^6x_4 + 2240x_1x_2^5x_3^2 - 2240x_1x_2^5x_4^2 + 10180x_1x_2^4x_3^3 \\
& - 23440x_1x_2^4x_3^2x_4 + 23440x_1x_2^4x_3x_4^2 - 10180x_1x_2^4x_4^3 + 10180x_1x_2^3x_3^4 - 10180x_1x_2^3x_4^4 \\
& + 2240x_1x_2^2x_3^5 + 23440x_1x_2^2x_3^4x_4 - 23440x_1x_2^2x_3^3x_4^2 - 2240x_1x_2^2x_3^2x_4^3 - 60x_1x_2x_3^6 \\
& + 23440x_1x_2x_3^5x_4^2 - 23440x_1x_2x_3^4x_4^3 + 60x_1x_2x_4^6 - 60x_1x_3^6x_4 - 2240x_1x_3^5x_4^2 \\
& + 10180x_1x_3^4x_4^3 - 10180x_1x_3^3x_4^4 + 2240x_1x_3^2x_4^5 + 60x_1x_3x_4^6 + x_2^8 - 6x_2^6x_3^2 \\
& - 60x_2^6x_3x_4 - 6x_2^6x_4^2 - 2240x_2^5x_3^2x_4 - 2240x_2^5x_3x_4^2 - 14x_2^4x_3^4 + 10180x_2^4x_3^3x_4 \\
& + 40412x_2^4x_3^2x_4^2 + 10180x_2^4x_3x_4^3 - 14x_2^4x_4^4 - 10180x_2^3x_3^4x_4 - 111980x_2^3x_3^3x_4^2 \\
& - 111980x_2^3x_3^2x_4^3 - 10180x_2^3x_3x_4^4 - 6x_2^2x_3^6 + 2240x_2^2x_3^5x_4 + 40412x_2^2x_3^4x_4^2 \\
& + 111980x_2^2x_3^3x_4^3 + 40412x_2^2x_3^2x_4^4 + 2240x_2^2x_3x_4^5 - 6x_2^2x_4^6 + 60x_2x_3^6x_4 \\
& - 2240x_2x_3^5x_4^2 - 10180x_2x_3^4x_4^3 - 10180x_2x_3^3x_4^4 - 2240x_2x_3^2x_4^5 + 60x_2x_3x_4^6 \\
& + x_3^8 - 6x_3^6x_4^2 - 14x_3^4x_4^4 - 6x_3^2x_4^6 + x_4^8.
\end{aligned}$$

This is the polynomial of Theorem 1. It has coefficients in \mathbb{Q} but note for instance that the singular points of $\mathcal{Z}(g)$, $\mathcal{Z}(g[2])$ and $\mathcal{Z}(g[3])$ have coordinates in various field extensions of \mathbb{Q} , and most of the singular points are not real (at least in this model). So, let ξ denote a primitive 24-th root of unity and let $\zeta = \xi^2$, $\zeta_4 = \zeta^3$ and $\zeta_3 = \zeta^4$, so that ζ , ζ_4 and ζ_3 are roots of unity of order 12, 4 and 3 respectively. Let α be a square root of

$$18\xi^6 + 14\xi^5 + 48\xi^4 + 2\xi^3 - 36\xi^2 - 14\xi - 24.$$

Let K denote the cyclotomic field $\mathbb{Q}(\zeta)$ and let $L = \mathbb{Q}(\zeta, \alpha)$: it is a Galois extension of \mathbb{Q} . We will check that the singular points of $\mathcal{Z}(g)$ and $\mathcal{Z}(g[3])$ are defined over K , while the singular points of $\mathcal{Z}(g[2])$ are defined over L .

Remark 1. As we will see in the computations below, the polynomial $g[3]$ is invariant under the action of the complex reflection group denoted by G_{32} in the Shephard–Todd classification [ShTo]. In fact, we did not find g directly: we found first $g[3]$ by looking at invariants of degree 24 of G_{32} , following ideas of Barth [Bar] and Sarti [Sar1], [Sar2], [Sar3] (for constructing the *Barth sextic* with 65 nodes and, for instance, the *Sarti surface* of degree 12 with 600 nodes), who used invariants of Coxeter groups of type H_3 and H_4 . ■

Remark 2. As the expert in computer algebra will notice by reading the MAGMA code below, the author is not a specialist in programming and asks for some indulgence. ■

An important feature of the polynomials g , $g[2]$ and $g[3]$ is that they are invariant under the action of various linear groups: this will be helpful to treat the big number of singularities, by separating the various orbits of singular points under their action. Let W_1 be the subgroup of $\mathbf{GL}_4(\mathbb{C})$ generated by

$$s_1 = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}, \quad s_2 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix} \quad \text{and} \quad s_3 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \end{pmatrix}.$$

Let W_2 be the subgroup of $\mathbf{GL}_4(\mathbb{C})$ generated by

$$t_1 = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & \zeta_4 \end{pmatrix}, \quad t_2 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & \zeta_4 \end{pmatrix} \quad \text{and} \quad t_3 = \begin{pmatrix} -\zeta_4 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}.$$

Finally, let W_3 denote the subgroup of $\mathbf{GL}_4(\mathbb{C})$ generated by

$$u_1 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \zeta_3 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \quad u_2 = \begin{pmatrix} \frac{\zeta_3+2}{3} & \frac{\zeta_3-1}{3} & \frac{\zeta_3-1}{3} & 0 \\ \frac{\zeta_3-1}{3} & \frac{\zeta_3+2}{3} & \frac{\zeta_3-1}{3} & 0 \\ \frac{\zeta_3-1}{3} & \frac{\zeta_3-1}{3} & \frac{\zeta_3+2}{3} & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix},$$

$$u_3 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & a & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} \quad \text{and} \quad u_4 = \begin{pmatrix} \frac{\zeta_3+2}{3} & \frac{1-\zeta_3}{3} & 0 & \frac{1-\zeta_3}{3} \\ \frac{1-\zeta_3}{3} & \frac{\zeta_3+2}{3} & 0 & \frac{\zeta_3-1}{3} \\ 0 & 0 & 1 & 0 \\ \frac{1-\zeta_3}{3} & \frac{\zeta_3-1}{3} & 0 & \frac{\zeta_3+2}{3} \end{pmatrix}.$$

Proposition 2. *If $1 \leq i \leq 3$, then $g[i]$ is invariant under the action of W_i .*

Commentaries. Let $Z(W_i)$ denote the center of W_i . In all cases, it is isomorphic to a group of roots of unity acting by scalar multiplication. Then:

- The group W_1 has order 48 and is isomorphic to the non-trivial double cover $\tilde{\mathfrak{S}}_4$ of the symmetric group \mathfrak{S}_4 . Hence $\text{Aut}(\mathcal{Z}(g))$ contains a subgroup isomorphic to \mathfrak{S}_4 .
- The group W_2 has order 768, contains a normal abelian subgroup H of order 32 and $W_2/H \simeq \mathfrak{S}_4$. The group $W_2/Z(W_2)$ has order 192, is contained in $\text{Aut}(\mathcal{Z}(g[2]))$, but is not isomorphic to a Coxeter group of type D_4 .
- The group W_3 is the complex reflection group denoted by G_{32} in the Shephard–Todd classification (it has order 155920). Recall that the group $W_3/Z(W_3)$, which is a subgroup of $\text{Aut}(\mathcal{Z}(g[3]))$, is a simple group of order 25920, is contained in $\text{Aut}(\mathcal{Z}(g[3]))$, and is isomorphic to the derived subgroup of the Weyl group of type E_6 (i.e. to the derived subgroup of the special orthogonal group $\mathbf{SO}_5(\mathbb{F}_3)$). Note that we have used the form implemented by Michel [Mic] in the Chevie package of Gap3.

It is natural to wonder whether $\text{Aut}(\mathcal{Z}(g[i]))$ is actually equal to $W_i/Z(W_i)$ for $i = 1, 2, 3$. ■

So let us start by defining the polynomial g and the three groups W_i in MAGMA and checking the facts stated in Proposition 2 and Commentaries. These data, together with the definition of the fields K and L as well as a function `ProjectiveOrbit` used for computing orbits of various points under the action of the groups W_i , are all contained in a file `g32-article.m`, whose content is given in the Appendix. Note that we will first work with the projective space over \mathbb{Q} , and the polynomial g will be defined over \mathbb{Q} .

```

> load 'g32-article.m';
>
> Order(W1);
48
> Order(Centre(W1));
2
> bool:=IsIsomorphic(W1/Centre(W1),SymmetricGroup(4));
> bool;
true
> Order(DerivedSubgroup(W1));
24
> gK:=ChangeRing(g,K);
> [gK^i eq gK : i in Generators(W1)];
[ true, true, true ]
>
> Order(W2);
768
> Order(Centre(W2));
4
> g2L:=ChangeRing(g2,L);
> [g2L^i eq g2L : i in Generators(W2)];
[ true, true, true ]
>
> Order(W3);
155520
> Order(W3/Centre(W3));
25920
> IsSimple(W3/Centre(W3));
true
> g3K:=ChangeRing(g3,K);
> [g3K^i eq g3K : i in Generators(W3)];
[ true, true, true, true ]

```

We now turn to the study of the singularities of the varieties $\mathcal{Z}(g[i])$ for $i \in \{1, 2, 3\}$. Note the following fact, that will be used further:

Lemma 3. *If $1 \leq i < j \leq 4$, then the closed subscheme of $\mathbf{P}^3(\mathbb{C})$ defined by the ideal $\langle g, \frac{\partial g}{\partial x_i}, \frac{\partial g}{\partial x_j} \rangle$ has dimension 0.*

This is checked thanks to the following code:

```

> dg:=[Derivative(g,i) : i in [1..4]];
> pairs:=[[1,2],[1,3],[1,4],[2,3],[2,4],[3,4]];
> time [Dimension(Scheme(P3,[g,dg[i[1]],dg[i[2]]))):
>   i in pairs];
[ 0, 0, 0, 0, 0, 0 ]
Time: 20.510 // about 20 seconds

```

1. Degree 8

Note that the fact that $\dim \mathcal{Z}(g)_{\text{sing}} = 0$ (see Lemma 3) implies that $\mathcal{Z}(g)$ is geometrically irreducible. Let \mathcal{U} denote the open subset of $\mathbf{P}^3(\mathbb{C})$ defined by $x_1 x_2 x_3 x_4 \neq 0$. We first check that $\mathcal{Z}(g)_{\text{sing}}$ is contained in \mathcal{U} .

```

> Zg:=Surface(P3,g);
> Zgsing:=SingularSubscheme(Zg);
> H:=Scheme(P3,x1*x2*x3*x4);
> Dimension(Intersection(Zgsing,H));
-1

```

In particular, all the singular points are contained (for instance) in the affine chart $\mathcal{Z}(g)^{\text{aff}}$ defined by “ $x_4 \neq 0$ ”. We will make all the remaining computations in this affine chart (and extend the scalars to the field K):

```

> Zgaff:=AffinePatch(Zg,1);
> ZgaffK:=ChangeRing(Zgaff,K);
> ZgaffKsing:=SingularSubscheme(ZgaffK);
> irr1:=IrreducibleComponents(ZgaffKsing);
> irr1:=[ReducedSubscheme(i) : i in irr1];
> Set([Degree(i) : i in irr1]);
{ 1 }
> ZgK:=ChangeRing(Zg,K);
> sings1:=[[Coordinates(i) : i in RationalPoints(j)] :
> j in irr1];
> sings1:=&cat sings1;
> sings1:=[ZgK ! (i cat [1]) : i in sings1];
> # sings1;
44

```


The result of the command `Set([Degree(i) : i in irr1])` shows that all singular points have coordinate in K , and the last command shows that the number of singular points in $\mathcal{Z}(g)$ is equal to 44. We then determine the W_1 -orbits in $\mathcal{Z}(g)_{\text{sing}}$ and check that they are all quotient singularities of type D_4 by picking up one point in each orbit.

```

> orbits:=[];
> test:=sings1;
> while (# test) gt 0 do
while> orb:=ProjectiveOrbit(W1,test[1]);
while> orb:=[ZgK ! Coordinates(i) : i in orb];
while> orbits:=orbits cat [orb];
while> test:=[i : i in test | (i in orb) eq false];
while> end while;
> [# i : i in orbits];
[ 24, 8, 12 ]
> for i in orbits do
for> print IsSimpleSurfaceSingularity(i[1]);
for> end for;
true D 4
true D 4
true D 4

```

This concludes the proof of statement (a) of Theorem 1. As a consequence of the above computations, the group W_1 does not act transitively on the set of singular points (it admits 3 orbits, of respective length 8, 12 and 24). Note that the points in the W_1 -orbit of cardinality 8 are the only real singular points of $\mathcal{Z}(g)$. Figure 1 shows part of the real locus of $\mathcal{Z}(g)$.

FIGURE 1. Part of the real locus of $\mathcal{Z}(g)$

2. Degree $8k$

Let \mathcal{U} denote the open subset of $\mathbf{P}^3(\mathbb{C})$ defined by $x_1 x_2 x_3 x_4 \neq 0$ and let $\sigma_k : \mathbf{P}^3(\mathbb{C}) \rightarrow \mathbf{P}^3(\mathbb{C})$, $[x_1; x_2; x_3; x_4] \mapsto [x_1^k; x_2^k; x_3^k; x_4^k]$. The restriction of σ_k to a morphism $\mathcal{U} \rightarrow \mathcal{U}$ is an étale Galois covering, with group $(\mu_k)^4 / \Delta \mu_k$ (here, μ_k denotes the group of k -th roots of unity in \mathbb{C}^\times and $\Delta : \mu_k \hookrightarrow (\mu_k)^4$ is the diagonal embedding). We have $\mathcal{Z}(g[k]) = \sigma_k^*(\mathcal{Z}(g))$.

Let us first prove that $\mathcal{Z}(g[k])$ is irreducible. We may assume that $k \geq 2$, as the result has been proved for $k = 1$ in the previous section. Recall that

$$\frac{\partial g[k]}{\partial x_i} = k x_i^{k-1} \left(\frac{\partial g}{\partial x_i} \circ \sigma_k \right),$$

so the singular locus of $\mathcal{Z}(g[k])$ is contained in

$$\{p_1, p_2, p_3, p_4\} \cup \left(\bigcup_{i \neq j} \sigma_k^*(\mathcal{Z}_{i,j}) \right),$$

where $p_i = [\delta_{i1}; \delta_{i2}; \delta_{i3}; \delta_{i4}]$ (and δ_{ij} is the Kronecker symbol) and $\mathcal{Z}_{i,j}$ is the subscheme of $\mathbf{P}^3(\mathbb{C})$ defined by the ideal $\langle g, \frac{\partial g}{\partial x_i}, \frac{\partial g}{\partial x_j} \rangle$ (and which has dimension 0 by Lemma 3). Since σ_k is finite, this implies that $\mathcal{Z}(g[k])_{\text{sing}}$ has dimension 0, so $\mathcal{Z}(g[k])$ is irreducible.

Now, as $\sigma_k : \mathcal{U} \rightarrow \mathcal{U}$ is étale and the singular locus of $\mathcal{Z}(g)$ is contained in \mathcal{U} (see the command `Dimension(Intersection(Zgsing, H))` of the previous section), the 44 singularities of $\mathcal{Z}(g)$ give $44k^3$ singularities in $\mathcal{Z}(g[k])$ of the same type, i.e. quotient singularities of type D_4 . This proves the statement (b) of Theorem 1.

Note that, for $k = 2, 3$ and 4 (and maybe for bigger k) we will prove in the next sections that $\mathcal{Z}(g[k])$ contains singular points outside of \mathcal{U} .

3. Degree 16

Using the morphism σ_2 defined in the previous section and the fact that it is étale over \mathcal{U} , we get that $\mathcal{Z}(g[2]) \cap \mathcal{U}$ has exactly 352 singular points, which are all quotient singularities of type D_4 . We now need to determine the singularities which are not contained in \mathcal{U} . So let \mathcal{H} be the complement of \mathcal{U} in $\mathbf{P}^3(\mathbb{C})$.

```

> Zg2:=Surface(P3,g2);
> Zg2sing:=SingularSubscheme(Zg2);
> H:=Scheme(P3,x1*x2*x3*x4);
> Zg2singH:=Intersection(Zg2sing,H);
> time Zg2singH:=ReducedSubscheme(Zg2singH);
Time: 11.300
> irr2H:=IrreducibleComponents(Zg2singH);
> # irr2H;
18
> Set([Degree(i) : i in irr2H]);
{ 4, 8 }
> &+ [Degree(i) : i in irr2H];
120

```

The last two commands show that $\mathcal{Z}(g)_{\text{sing}} \cap \mathcal{H}$ contains 120 points. We now check that all the singular points contained in \mathcal{H} have coordinates in L :

```

> Zg2L:=ChangeRing(Zg2,L);
> sings2H:=[[Coordinates(i) :
>   i in RationalPoints(ChangeRing(j,L))] :
>   j in irr2H];
> sings2H:=&cat sings2H;
> sings2H:=[Zg2L ! i : i in sings2H];
> # sings2H;
120

```

We now determine the W_2 -orbits in $\mathcal{Z}(g)_{\text{sing}} \cap \mathcal{H}$: there is one W_2 -orbit of cardinality 24 (and we check that its elements are quotient singularities of type A_1) and one of cardinality 96.

```


> orbits2:=[];
> test:=sings2H;
> W2L:=ChangeRing(W2,L);
> while (# test) gt 0 do
while>   orb:=ProjectiveOrbit(W2L,test[1]);
while>   orb:=[Zg2L ! Coordinates(i) : i in orb];
while>   orbits2:=orbits2 cat [orb];
while>   test:=[i : i in test | (i in orb) eq false];
while> end while;
> [# i : i in orbits2];
[ 24, 96 ]
> IsSimpleSurfaceSingularity(orbits2[1][1]);
true A 1

```

We now study the singularity of $\mathcal{Z}(g[2])$ at the points of the orbit of cardinality 96. It turns out that the command `IsSimpleSurfaceSingularity` takes too much time to get a conclusion, so we will investigate properties of the equation of $\mathcal{Z}(g[2])$ in a neighborhood of the first point p (in Magma list `orbits[2]`). We work in the affine chart $x_3 \neq 0$ (where $p = [\xi_1; \xi_2; \xi_3; \xi_4]$ lives), and we denote by (x, y, z) the coordinates of the affine chart $x_3 \neq 0$ equal to $(x_1/x_3 + \xi_1/\xi_3, x_2/x_3 + \xi_2/\xi_3, x_4/x_3 + \xi_4/\xi_3)$ and we set

$$f(x, y, z) = g[2](x, y, 1, z).$$

If $j \geq 0$, we denote by f_j the homogeneous component of f of degree j . As $p \in \mathcal{Z}(g[2])_{\text{sing}}$, we have $f_0 = f_1 = 0$.

FIGURE 2. Part of the real locus of $\mathcal{Z}(g[2])$

```

> p:=orbits2[2][1];
> A3L<x,y,z>:=AffineSpace(L,3);
> cop:=Coordinates(p);
> f:=Evaluate(g2L,[x+cop[1],y+cop[2],1,z+cop[4]]);
> cof:=Coefficients(f);
> mof:=Monomials(f);
> l:=# mof;
> f2:=&+ [cof[i]*mof[i] : i in [1..l] |
>   Degree(mof[i]) eq 2];
> f2:=f2/Reverse(Coefficients(f2))[1];
> f2;
> Factorization(f2-z^2);
[
  <x + (3*xi^7 + 4*xi^5 + xi^3 - xi)*y, 2>
]

```

The last command shows that there exists a linear change of the coordinates $(x, y, z) \mapsto (X, Y, Z)$ such that f_2 might be transformed into $X^2 + Z^2$. By standard arguments, this proves that p is a quotient singularity of type A_k , for some $k \geq 2$, which can be obtained as the Milnor number of f : however, Magma cannot compute this Milnor number in a reasonable amount of time and we need to copy the polynomial f in the software Singular [DGPS] to compute this Milnor number (!): we obtain 2. So p is a quotient singularity of type A_2 . This completes the proof of statement (c) of Theorem 1.

Figure 2 shows part of the real locus of $\mathcal{Z}(g)$.


4. Degree 24

We now come to the surface $\mathcal{Z}(g[3])$. From Section 2, we know that $\mathcal{Z}(g[3])_{\text{sing}} \cap \mathcal{U}$ contains $44 \times 3^3 = 1188$ singular points, which are all quotient singularities of type D_4 . Let us compute $\mathcal{Z}(g[3])_{\text{sing}} \cap \mathcal{H}$:

```

> Zg3:=Surface(P3,g3);
> Zg3sing:=SingularSubscheme(Zg3);
> Zg3singH:=Intersection(Zg3sing,H);

```

FIGURE 3. Part of the real locus of $\mathcal{Z}(g[3])$

```

> time Zg3singH:=ReducedSubscheme(Zg3singH);
Time: 19.320
> time irr3H:=IrreducibleComponents(Zg3singH);
Time: 18.170
> # irr3H;
72
> Set([Degree(i) : i in irr3H]);
{ 2, 4 }
> &+ [Degree(i) : i in irr3H];
252

```

The last command shows that $\mathcal{Z}(g[3])_{\text{sing}} \cap \mathcal{H}$ contains 252 points. We now show that they are all defined over K :

```

> Zg3K:=ChangeRing(Zg3,K);
> sings3H:=[[Coordinates(i) :
>   i in RationalPoints(ChangeRing(j,K))] :
>   j in irr3H];
> sings3H:=&cat sings3H;
> sings3H:=[Zg3K ! i : i in sings3H];
> # sings3H;
252

```

So $\mathcal{Z}(g[3])_{\text{sing}}$ contains 1440 points, and we now check that W_3 acts transitively on them:

```

> p:=sings3H[1];
> time # ProjectiveOrbit(W3,p);
1440
Time: 29.850

```

The proof of statement (d) of Theorem 1 is complete. Note also that, in the given model, the surface $\mathcal{Z}(g[3])$ has only 32 real singular points: Figure 3 gives partial views of its real locus.

Remark 3. From Section 2, we deduce that $\mathcal{Z}(g[4])_{\text{sing}}$ has 2816 quotient singularities of type D_4 lying in the open subset \mathcal{U} and it can be checked that it has 480 other singular points lying in \mathcal{H} , for which we did not determine the type. ■

Remark 4. After investigations in the invariant rings of several complex reflection groups, we have also been able to construct curves with many singularities. For example:

- Using the reflection group $W = G_{24}$, we have obtained a cuspidal curve of degree 14 in $\mathbf{P}^2(\mathbb{C})$ with exactly 42 cusps (all lying in a single W -orbit). Note that $W/Z(W)$ has order 168 and is isomorphic to $\mathbf{GL}_3(\mathbb{F}_2)$.
- Using the reflection group $W = G_{26}$, we have obtained a curve of degree 18 in $\mathbf{P}^2(\mathbb{C})$ with 72 cusps and 12 nodes (these are the two W -orbits of singular points). Note that $W/Z(W)$ has order 216.

Also, other singular surfaces have been obtained. For example:

- Using the reflection group $W = G_{29}$ (note that $W/Z(W)$ has order 1920), we have obtained:
 - a surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with 160 nodes, all belonging to the same W -orbit. Recall that the *Endraß octic* [End] has degree 8 and 168 nodes (and its automorphism group has order 16) while the *Sarti octic* [Sar1] has 144 nodes.
 - a surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with 20 singular points of multiplicity 3 and Milnor number 11, all belonging to the same W -orbit.
- Using the reflection group $W = G_{31}$ (note that $W/Z(W)$ has order 11520), we have obtained a surface of degree 20 in $\mathbf{P}^3(\mathbb{C})$ with 1920 nodes, all lying in the same W -orbit. Recall that the *Chmutov surface* [Chm] of degree 20 has 2926 nodes.

Details will appear in a forthcoming paper. ■

Acknowledgements. This paper is based upon work supported by the National Science Foundation under Grant No. DMS-1440140 while the author was in residence at the Mathematical Sciences Research Institute in Berkeley, California, during the Spring 2018 semester. The hidden computations which led to the discovery of the polynomial g were done using the High Performance Computing facilities of the MSRI.

I wish to thank warmly Alessandra Sarti, Oliver Labs and Duco van Straten for useful comments and references and Gunter Malle for a careful reading of a first version of this note. Figures were realized using the software SURFER [Sur].

Appendix

This appendix gives a copy of the file `g32-article.m` loaded at the beginning of the computations. It contains the data of the polynomial g , the fields K and L , the three groups W_i and the function `ProjectiveOrbit` which is used throughout the computations (it is certainly not the most efficient code, but it is sufficient for our purpose). Note that W_1 and W_3 are defined over the field K , while W_2 is defined over the field L .

```

Q:=RationalField();
P3<x1,x2,x3,x4>:=ProjectiveSpace(Q,3);
K<zeta>:=CyclotomicField(12);
zeta3:=zeta^4;
P3K:=ProjectiveSpace(K,3);
K24<xi>:=CyclotomicField(24);
zeta4:=xi^6;
POL<T>:=PolynomialRing(K24);
L<alpha>:=NumberField(T^2-(18*xi^6 + 14*xi^5
+ 48*xi^4 + 2*xi^3 - 36*xi^2 - 14*xi - 24));
P3L:=ProjectiveSpace(L,3);

```

```

g:=x1^8 - 6*x1^6*x2^2 - 60*x1^6*x2*x3
- 60*x1^6*x2*x4 - 6*x1^6*x3^2 + 60*x1^6*x3*x4
- 6*x1^6*x4^2 + 2240*x1^5*x2^2*x3
- 2240*x1^5*x2^2*x4 + 2240*x1^5*x2*x3^2
- 2240*x1^5*x2*x4^2 + 2240*x1^5*x3^2*x4
- 2240*x1^5*x3*x4^2 - 14*x1^4*x2^4
+ 10180*x1^4*x2^3*x3 + 10180*x1^4*x2^3*x4
+ 40412*x1^4*x2^2*x3^2 - 23440*x1^4*x2^2*x3*x4
+ 40412*x1^4*x2^2*x4^2 + 10180*x1^4*x2*x3^3
+ 23440*x1^4*x2*x3^2*x4 + 23440*x1^4*x2*x3*x4^2
+ 10180*x1^4*x2*x4^3 - 14*x1^4*x3^4
- 10180*x1^4*x3^3*x4 + 40412*x1^4*x3^2*x4^2
- 10180*x1^4*x3*x4^3 - 14*x1^4*x4^4
+ 10180*x1^3*x2^4*x3 - 10180*x1^3*x2^4*x4
+ 111980*x1^3*x2^3*x3^2 - 111980*x1^3*x2^3*x4^2
+ 111980*x1^3*x2^2*x3^3 - 111980*x1^3*x2^2*x4^3
+ 10180*x1^3*x2*x3^4 - 10180*x1^3*x2*x4^4
+ 10180*x1^3*x3^4*x4 - 111980*x1^3*x3^3*x4^2
+ 111980*x1^3*x3^2*x4^3 - 10180*x1^3*x3*x4^4
- 6*x1^2*x2^6 + 2240*x1^2*x2^5*x3 + 2240*x1^2*x2^5*x4
+ 40412*x1^2*x2^4*x3^2 + 23440*x1^2*x2^4*x3*x4
+ 40412*x1^2*x2^4*x4^2 + 111980*x1^2*x2^3*x3^3
+ 111980*x1^2*x2^3*x4^3 + 40412*x1^2*x2^2*x3^4
+ 154704*x1^2*x2^2*x3^2*x4^2 + 40412*x1^2*x2^2*x4^4
+ 2240*x1^2*x2*x3^5 - 23440*x1^2*x2*x3^4*x4
- 23440*x1^2*x2*x3*x4^4 + 2240*x1^2*x2*x4^5
- 6*x1^2*x3^6 - 2240*x1^2*x3^5*x4
+ 40412*x1^2*x3^4*x4^2 - 111980*x1^2*x3^3*x4^3
+ 40412*x1^2*x3^2*x4^4 - 2240*x1^2*x3*x4^5
- 6*x1^2*x4^6 - 60*x1*x2^6*x3 + 60*x1*x2^6*x4
+ 2240*x1*x2^5*x3^2 - 2240*x1*x2^5*x4^2
+ 10180*x1*x2^4*x3^3 - 23440*x1*x2^4*x3^2*x4
+ 23440*x1*x2^4*x3*x4^2 - 10180*x1*x2^4*x4^3
+ 10180*x1*x2^3*x3^4 - 10180*x1*x2^3*x4^4
+ 2240*x1*x2^2*x3^5 + 23440*x1*x2^2*x3^4*x4
- 23440*x1*x2^2*x3*x4^4 - 2240*x1*x2^2*x4^5
- 60*x1*x2*x3^6 + 23440*x1*x2*x3^4*x4^2
- 23440*x1*x2*x3^2*x4^4 + 60*x1*x2*x4^6 - 60*x1*x3^6*x4
- 2240*x1*x3^5*x4^2 + 10180*x1*x3^4*x4^3
- 10180*x1*x3^3*x4^4 + 2240*x1*x3^2*x4^5
+ 60*x1*x3*x4^6 + x2^8 - 6*x2^6*x3^2 - 60*x2^6*x3*x4
- 6*x2^6*x4^2 - 2240*x2^5*x3^2*x4 - 2240*x2^5*x3*x4^2
- 14*x2^4*x3^4 + 10180*x2^4*x3^3*x4
+ 40412*x2^4*x3^2*x4^2 + 10180*x2^4*x3*x4^3
- 14*x2^4*x4^4 - 10180*x2^3*x3^4*x4
- 111980*x2^3*x3^3*x4^2 - 111980*x2^3*x3^2*x4^3
- 10180*x2^3*x3*x4^4 - 6*x2^2*x3^6 + 2240*x2^2*x3^5*x4
+ 40412*x2^2*x3^4*x4^2 + 111980*x2^2*x3^3*x4^3
+ 40412*x2^2*x3^2*x4^4 + 2240*x2^2*x3*x4^5 - 6*x2^2*x4^6
+ 60*x2*x3^6*x4 - 2240*x2*x3^5*x4^2 - 10180*x2*x3^4*x4^3
- 10180*x2*x3^3*x4^4 - 2240*x2*x3^2*x4^5 + 60*x2*x3*x4^6
+ x3^8 - 6*x3^6*x4^2 - 14*x3^4*x4^4 - 6*x3^2*x4^6 + x4^8;
g2:=Evaluate(g, [x1^2, x2^2, x3^2, x4^2]);
g3:=Evaluate(g, [x1^3, x2^3, x3^3, x4^3]);

```

```

s1:=Matrix(K,4,4,
[[ 0, 1, 0, 0],
 [ 1, 0, 0, 0],
 [ 0, 0, 1, 0],
 [ 0, 0, 0, -1]]);
s2:=Matrix(K,4,4,
[[ 1, 0, 0, 0],
 [ 0, 0, 1, 0],
 [ 0, 1, 0, 0],
 [ 0, 0, 0, -1]]);
s3:=Matrix(K,4,4,
[[ 1, 0, 0, 0],
 [ 0, -1, 0, 0],
 [ 0, 0, 0, -1],
 [ 0, 0, -1, 0]]);
W1:=MatrixGroup<4,K | [s1,s2,s3]>;

t1:=Matrix(L,4,4,
[[ 0, 1, 0, 0],
 [ 1, 0, 0, 0],
 [ 0, 0, 1, 0],
 [ 0, 0, 0, zeta4]]);
t2:=Matrix(L,4,4,
[[ 1, 0, 0, 0],
 [ 0, 0, 1, 0],
 [ 0, 1, 0, 0],
 [ 0, 0, 0, zeta4]]);
t3:=Matrix(L,4,4,
[[-zeta4, 0, 0, 0],
 [ 0, 1, 0, 0],
 [ 0, 0, 0, 1],
 [ 0, 0, 1, 0]]);
W2:=MatrixGroup<4,L | [t1,t2,t3]>;

u1:=Matrix(K,4,4,
[ [ 1, 0, 0, 0 ],
 [ 0, 1, 0, 0 ],
 [ 0, 0, zeta3, 0 ],
 [ 0, 0, 0, 1 ]]);
u2:=Matrix(K,4,4,
[ [(zeta3+2)/3, (zeta3-1)/3, (zeta3-1)/3, 0 ],
 [(zeta3-1)/3, (zeta3+2)/3, (zeta3-1)/3, 0 ],
 [(zeta3-1)/3, (zeta3-1)/3, (zeta3+2)/3, 0 ],
 [ 0, 0, 0, 1 ]]);
u3:=Matrix(K,4,4,
[ [ 1, 0, 0, 0 ],
 [ 0, zeta3, 0, 0 ],
 [ 0, 0, 1, 0 ],
 [ 0, 0, 0, 1 ]]);
u4:=Matrix(K,4,4,
[ [(zeta3+2)/3, (1-zeta3)/3, 0, (1-zeta3)/3 ],
 [(1-zeta3)/3, (zeta3+2)/3, 0, (zeta3-1)/3 ],
 [ 0, 0, 1, 0 ],
 [(1-zeta3)/3, (zeta3-1)/3, 0, (zeta3+2)/3 ]]);

```

```

W3:=MatrixGroup<4,K | [u1,u2,u3,u4]>;

// ProjectiveOrbit computes orbit of
//points in projective space

ProjectiveOrbit:=function(grp,pt)
  local i,j,res,v,w,V,PROJ,grpmod,zgrp;
  zgrp:=Centre(grp);
  zgr:=[w : w in zgrp | IsScalar(w)];
  zgrp:=sub<grp | zgrp>;
  grpmod:=Transversal(grp,zgrp);
  V:=VectorSpace(grp);
  PROJ:=AmbientSpace(Scheme(pt));
  v:=V ! Coordinates(pt);
  res:=[PROJ ! Coordinates(V,v*Transpose(w)) : w in grp];
  return [i : i in Set(res)];
end function;

```

References

- [Bar] W. BARTH, *Two projective surfaces with many nodes, admitting the symmetries of the icosahedron*, J. Algebraic Geom. **5** (1996), 173–186.
- [Chm] S. V. CHMUTOV, *Examples of projective surfaces with many singularities*, J. Algebraic Geom. **1** (1992), 191–196.
- [DGPS] W. DECKER, G.-M. GREUEL, G. PFISTER & H. SCHÖNEMANN, *Singular 4-1-1 — A computer algebra system for polynomial computations*, <http://www.singular.uni-kl.de> (2018).
- [End] S. ENDRAß, *A projective surface of degree eight with 168 nodes*, J. Algebraic Geom. **6** (1997), 325–334.
- [Magma] W. BOSMA, J. CANNON & C. PLAYOUST, *The Magma algebra system. I. The user language*, J. Symbolic Comput. **24** (1997), 235–265.
- [Mic] J. MICHEL, *The development version of the CHEVIE package of GAP3*, J. of Algebra **435** (2015), 308–336.
- [Miy] Y. MIYAOKA, *The maximal number of quotient singularities on surfaces with given numerical invariants*, Math. Ann. **268** (1984), 159–171.
- [Sar1] A. SARTI, *Pencils of symmetric surfaces in \mathbf{P}^3* , J. of Algebra **246** (2001), 429–452.
- [Sar2] A. SARTI, *Symmetric surfaces with many singularities*, Comm. in Alg. **32** (2004), 3745–3770.
- [Sar3] A. SARTI, *Symmetrische Flächen mit gewöhnlichen Doppelpunkten*, Math. Semesterber. **55** (2008), 1–5.
- [ShTo] G.C. SHEPHARD & J.A. TODD, *Finite unitary reflection groups*, Canad. J. Math. **6** (1954), 274–304.
- [Sur] www.imaginary.org/program/surfer.

May 3, 2018

CÉDRIC BONNAFÉ, Institut Montpellierain Alexander Grothendieck (CNRS: UMR 5149), Université Montpellier 2, Case Courrier 051, Place Eugène Bataillon, 34095 MONTPELLIER Cedex, FRANCE
E-mail: cedric.bonnafe@umontpellier.fr